CFB Software

Astrobe

Oberon for Arm Cortex-M0, M3, M4 and M7 Microcontrollers

Astrobe
Oberon for Arm Cortex-M0, M3, M4 and M7 Microcontrollers

Table of Contents

1 Introduction

Astrobe is a fast and responsive integrated development environment for Windows. This can be used to write software to run on the powerful Arm Cortex-M0, M3, M4 and M7 microcontrollers. In the following when we refer to Astrobe for Cortex-M it applies to all of these versions.

Refer to the Astrobe website at https://www.astrobe.com/ for the latest information on the availability of the different versions of Astrobe.

2 File Descriptions

The Astrobe compiler and linker expect there to be a correspondence between the names of modules in the source code and the associated filenames.

When you are creating a new source code file you should give the file the same name as its module name with a *.mod* extension.

The filenames of module-related files created by Astrobe are made from the name of the module and one of the following file extensions:

Ext	Type	Created by	Used by	Scope	Description
.arm	Binary	Compile	Link	Module	Linkable object file
.asm	Text	Disassemble		Application	Disassembler listing
.bin	Binary	Link	Upload	Application	Linked binary executable file
.def	Text		Edit	Module	SYSTEM interface
.drf	Binary	Link	Disassemble Application	Application	Reference information
.ini	Text	Configuration	Compile Link Upload	Application	Compile, link, build and upload options
.lst	Text	Disassemble		Module	Disassembler listing
.map	Text	Link		Application	Code and data memory usage
.mod	Text	Edit	Compile	Module	Source code
.ref	Binary	Link	Traps	Application	Trap reference resource data
.res	Any		Link	Module	Resource data
.s	Text	Disassemble		Application	Assembler source
.smb	Binary	Compile	Compile	Module	Symbol file of exported items



2.1 Example

A module named *LcdDisplay* is saved as the file *LcdDisplay.mod*. When it is compiled the compiler generates a symbol file *LcdDisplay.smb* and an object file *LcdDisplay.arm*.

The main module of the application called *DigiClock* is saved as *DigiClock.mod*. *DigiClock* imports *LcdDisplay*.

When you are editing *DigiClock.mod* in the Astrobe editor you can automatically open the source code of *LcdDisplay* by clicking on its name in the IDE's Import navigation pane.

When *DigiClock* is compiled the compiler uses the information in the symbol file *LcdDisplay.smb* to ensure that the use of all of the variables, procedures etc. from *LcdDisplay* conforms to the declarations of those items in *LcdDisplay*. It is not necessary to have the source code of *LcdDisplay* available to validate the use of its exported items.

When *DigiClock* is linked the linker uses the Link Options data from the current configuration and combines the object files *Main.arm, DigiClock.arm*, *LcdDisplay.arm* and all other

imported modules. The linker creates the memory usage map file *DigiClock.map*, the trap reference resource file *DigiClock.ref* and the executable file *DigiClock.bin.*

When *DigiClock* is uploaded the flash memory of the target processor is programmed with the contents of the executable file.

2.2 Linking and Loading

An application created with Astrobe is made up from a selection of the following modules:

- *System Modules*

  Startup code module
  Astrobe MCU-specific library modules
  Astrobe general library modules

- *User-developed Modules*

  Common user library modules
  Application-specific modules
  Main module

The simplest application consists of a single Main module accessing the System Modules.

The Linker / Loader combines all of the components needed by an application into a single file in binary format suitable to be uploaded by Astrobe and executed on the target processor.

A feature of the Oberon language is that all of the information regarding dependencies between the various modules is defined in the source code. There is no need to create and maintain separate 'make files' as commonly used in other systems.

The only details the Astrobe Linker / Loader needs to know to be able to build an application are:

- The name of the main module
- The physical locations of the folders containing the library modules
- The start and end addresses of the data and code areas

When the Astrobe *Project > Link* command is selected the current module whose source code is in view is taken to be the main module.

The details of the code and data address ranges and the physical locations of the library files are as specified for the current *configuration*. See *Library Organisation* below for details.

If you are using the built-in function NEW to allocate memory from the ‘heap’ to dynamic POINTER variables you can also use the configuration feature to specify:

- The address of the start of the heap
- The limit of the heap

If you keep the default values the CPU RAM is shared between global variables, the stack (local variables) and the heap (POINTER variables). This is suitable for typical applications.

However, if your system has non-CPU RAM that is directly addressable in the same way as CPU RAM then you can change these values so the non-CPU RAM is used by the heap. More memory is then available for global and local variables.

The values entered are listed in the linker progress report and linker map file.

2.3 Startup Code

The stack pointer, interrupt vectors etc. are initialised by startup code generated by the linker. The startup code is the first part of the application to execute when the microcontroller is reset.

The initialisation code of each module of the application is then executed in turn starting with the lowest module in the dependency chain. Execution continues all the way up until the initialisation code of the main module is started and the application proceeds.

Memory mapping control and phase-locked loop (PLL) options of the microcontroller are configured in the process of initialising the Astrobe library module *Main*. The module *Main* must be included in the IMPORT list of the main module of every Astrobe application to ensure that the application is correctly initialised.

You can modify the source code of the *Main*, *MCU* and *Traps* modules to allow different configurations of memory mapping and PLL features and to customise the output of runtime error messages.

2.4 Library Folders

Groups of common files that are shared between several applications developed using Astrobe may be conveniently organised in a system of *library folders* avoiding the need to duplicate copies of common / shared files.

The following is an example of the Configurations, Library and Examples folder structure supplied with Astrobe for Cortex-M0:



The folder *Astrobe\Lib\General* contains the generic system library files (e.g. *Out.*, Reals.** etc.) that are common to all Cortex-M microcontrollers.

The remaining folders in *Astrobe\Lib* contain microcontroller-specific versions of the library files e.g. *Main.*, MCU.** etc.

The library folders are standard Windows folders containing collections of source (**.mod*), symbol (**.smb*) and object files (**.arm*).

2.5 Configuration Files

The Compile, Link, Build and Upload options for Cortex-M target microcontrollers are stored in *Configuration* (**.ini*) files. Examples of these are included with Astrobe for the target microcontrollers used on the supported development boards.

The commands on the Astrobe *Configuration* menu are used to maintain and access the configuration files. See the *Configuration Files* section of the Astrobe Help file for more information.

Configuration entries include the locations of the library folders and the code and data address ranges to be used when linking.

2.5.1 Library Folders

The list of library folders to be searched is stored in the configuration file. The name of each library folder is stored on a separate line in the configuration's *Library Pathnames* textbox. Examples are:

1. Astrobe for Cortex-M0: The target microcontroller is an STM32F091.

   D:\AstrobeM0-v9.3\Lib\STM32F091
   D:\AstrobeM0-v9.3\Lib\General

2. Astrobe for Cortex-M4: The target is microcontroller is an STM32F303ZE

   %AstrobeM4%\Lib\STM32F303ZE
   %AstrobeM4%\Lib\General

   where *%AstrobeM4%* is substituted with the location of the library and example files that you specified when you installed or last upgraded Astrobe for Cortex-M4.

The editor, compiler, linker and builder first search the <*current folder*> when trying to locate imported symbol and object files. They then search each of the library folders in the list. The search continues until the file is found or the last folder in the list has been searched.

<*current folder*> is the folder which contains the source file (**.mod*) currently being compiled or the main object file (**.arm*) currently being linked.

2.5.2 Data Addresses

The configuration files have entries, *Data Range* and *Code Range* to allow you to specify the Code and Data Flash and RAM address ranges to use when the Astrobe linker produces the binary executable file.

Developers targeting other MCUs can create new configuration files and develop their own hardware-specific library modules using the files and source code supplied with Astrobe as examples.

2.6 Uploading Executable Files

Development boards supported with Astrobe allow executable files (**.bin*) which were created by the Astrobe *Link* or *Build* commands to be uploaded via a USB connection from the PC to the development board. This is done using the Astrobe *Upload* command.

Otherwise 3rd-party tools (e.g. STM32 ST-Link) can be used to upload the executables. Developers can add entries to the Astrobe Tools menu to call command-line versions of these tools from within Astrobe. For example, to launch ST-Link to upload your current application to an STM32 microcontroller, you can add these details to your *Tools.ini* file:

```
[Command3]
MenuItem=&ST-LINK Upload
Path=C:\Program Files (x86)\STMicroelectronics\STM32 ST-LINK Utility\ST-LINK Utility\ST-
LINK_CLI.exe
Parameters=-P %FileRoot%.bin 0x08000000 -V -Q
WorkingFolder=%FileDir%
ConsoleApp=TRUE
```

See the *Tools Menu* section of the Astrobe Help file for a specification of each entry, and the ST-LINK documentation for a specification of the ST-LINK parameters.

2.7 Resource Data

The usual way to process constant data in an Oberon program is to declare the values in a CONST list or store them in a global array in the initialisation section of a module. Neither of these methods is practical when dealing with large amounts of constant data (e.g. the definition of a font, a bitmap image etc.).

Typically on a PC system, this sort of data would be stored in a file to be read at runtime. As a file system is often not available on the smaller embedded systems targeted by Astrobe, a different approach is required. The solution used is to gather together all of the relevant data files at link time and append them to the linked executable to be stored in Flash ROM when the program is uploaded.

A library module *ResData* is provided to allow the programmer to conveniently access the data from Flash ROM within the program as if it were data stored in a random-access disk file. A description of the available functions is included in the *Library Modules* section of this document.

Several resources can be attached to the one program; each is identified by its module name. Typically, the steps involved in making a resource file are:

- Make a copy of the original data file
- Rename the copy to match the associated module name with the extension *.res*
- Move the renamed copy to the folder which contains the source code of the module

At link time, after the Astrobe linker has linked all of the object files *<module>.arm* into the executable program, it looks for the corresponding resource files named *<module>.res* and appends them to the executable.

If you need to associate several different resource files with one module you could create an empty resource module for each separate resource e.g.

```
MODULE MyData;
END MyData.
```

and then include the names of those resource modules in the IMPORT list of the associated module.

The resource file can contain any type of data. How that data is interpreted is determined by the programmer. The only requirement is that the size of the file is a multiple of four bytes.

Study the source code of the *SlideDemo* application included with Astrobe for an example of how to use resource files. The application reads AutoCad Slide-format vector images from resource files and displays them on a screen.

3 Library Modules

The following library modules are included with Astrobe for Cortex-M:

Module name	Description
Bits	Bitwise operations on integers
Clock	RTC time
Convert	Conversion of integers to / from strings
DateTime	Date and time formatting
Error	Error messages referenced by Traps
FPU (M0, M3)	Support of mathematical operations on floating point numbers
GPIO	General Purpose IO pin configuration and control
Graphics	Draw lines, circles and ellipses on display devices
HCDrive	HCFiler - Low-level sector functions for SDHC Cards
HCDir	HCFiler - Directory functions
HCFiles	HCFiler - File output functions
I2C	Reading from and writing to the I2C bus in Master mode
In	Formatted ASCII text input
LinkOptions	Values of options supplied by the user at link time
Main	Initialisation code required by an application
Math	Basic mathematical and trigonometric functions
MAU	Memory allocation unit
MCU	Microcontroller-specific definitions and peripheral addresses
Out	Formatted ASCII text output
Put	String-handling helper functions used by *Convert* and *Reals*
Random	Pseudo-random number generator
Reals	Real number support and conversion to / from strings
ResData	Access constant user data attached to the program by the linker
RTC	Real-time Clock time and date
Serial	Basic polled UART serial IO
SPI	Reading from and writing to the Serial Peripheral Interface bus
Storage	User-definable memory allocation / deallocation procedures
Strings	General string-handling functions
SYSTEM	Implementation-specific low level functions
Time	Time formatting
Timers	Microsecond and millisecond time measurement and delays
Traps	Runtime error trapping

FPU, MAU and *SYSTEM* are special i.e. they are dependent on the version of the compiler and must follow some specific conventions. See the library module descriptions below for further details.

If a user module calls the Oberon built-in functions *NEW* or *DISPOSE* then module *MAU* is called and the *MAU* module is automatically imported. *MAU* also contains some functions that can be called explicitly by a user module. The programmer must then include *MAU* in the module's IMPORT list.

FPU is only needed for Astrobe for Cortex-M0 and M3. If a user module uses mathematical operations (e.g. divide, multiply etc.) on variables that are declared as REALs then an *FPU* function is called and the *FPU* module automatically imported. Normally a user module would not explicitly call an *FPU* function.

All other library modules are normal i.e.

- They must be explicitly imported by modules which access their exported items.
- They could be replaced with alternative versions developed by an Astrobe user.

Some library procedures use assertions to check that the values of input parameters are within a valid range. Invalid values result in a runtime assertion error. The error codes and reason for the error are listed in the section titled *Runtime Error Codes* below.

3.1 Bits

Definition

```
DEFINITION MODULE Bits;

PROCEDURE* And*(i, j: INTEGER): INTEGER;

PROCEDURE* Or*(i, j: INTEGER): INTEGER;

PROCEDURE* Not*(i: INTEGER): INTEGER;

PROCEDURE* Nand*(i, j: INTEGER): INTEGER;

PROCEDURE* Nor*(i, j: INTEGER): INTEGER;

PROCEDURE* Xor*(i, j: INTEGER): INTEGER;

PROCEDURE* Xnor*(i, j: INTEGER): INTEGER;

END Bits.
```

Description

The module Bits contains functions for performing bitwise operations on integer values. For example, if:

```
i := 0FF00 FF00H;
j := 0F0F0 0F0FH;
```

then:

```
And(i, j)  = 0F000 0F00H

Or(i, j)   = 0FFF0 FF0FH

Not(j)     = 000FF 00FFH

Nand(i, j) = 00FFF F0FFH

Nor(i, j)  = 0000F 00F0H

Xor(i, j)  = 00FF0 F00FH

Xnor(i, j) = 0F00F 0FF0H
```

3.2 Clock

Definition

```
DEFINITION MODULE Clock;

IMPORT MCU, SYSTEM;

PROCEDURE* Pack*(hh, mm, ss: INTEGER; VAR ctime: INTEGER);

PROCEDURE* Unpack*(ctime: INTEGER; VAR hh, mm, ss: INTEGER);

PROCEDURE* Time*(): INTEGER;

PROCEDURE* GetHMS*(VAR hh, mm, ss: INTEGER);

PROCEDURE* SetHMS*(hh, mm, ss: INTEGER);

PROCEDURE* Hours*(): INTEGER;

PROCEDURE* Minutes*(): INTEGER;

PROCEDURE* Seconds*(): INTEGER;

PROCEDURE* Init*();

END Clock.
```

Description

The module Clock contains functions for accessing the time components of the Real Time Clock (RTC).

Init should be called before any other Clock procedures are called. Using the current value of PCLK it calculates and loads the values of the Prescaler integer and fraction registers, resets the clock and then enables it.

SetHMS disables the clock, updates the value in the Consolidated Time Register 0 using the hours, minutes and seconds parameters with the day of week (DOW) set to zero. The clock is then re-enabled.

Time returns the value read from the Consolidated Time Register 0 with the day of week (DOW) set to zero. The individual hours, minutes and seconds fields can be extracted from this value using the *Unpack* function.

GetHMS is equivalent to:

```
Unpack(Time(), hh, mm, ss)
```

Hours, *Minutes* and *Seconds* can be used to conveniently access the individual components of the time if they are not all required.

3.3 Convert

Definition

```
DEFINITION MODULE Convert;

IMPORT Error, Put;

CONST
  (* possible values for result *)
  noError* = 0;
  overflow* = 1;
  syntaxError* = 2;

PROCEDURE StrToInt*(str: ARRAY OF CHAR; VAR n: INTEGER; VAR result: INTEGER);

PROCEDURE IntToStr*(n: INTEGER; VAR s: ARRAY OF CHAR);

PROCEDURE IntToHex*(n: INTEGER; VAR s: ARRAY OF CHAR);

END Convert.
```

Description

The module Convert contains functions for converting integers to strings and vice versa.

3.4 DateTime

Definition

```
DEFINITION MODULE DateTime;

IMPORT Put, RTC;

CONST
  Jan* = 1;
  Feb* = 2;
  Mar* = 3;
  Apr* = 4;
  May* = 5;
  Jun* = 6;
  Jul* = 7;
  Aug* = 8;
  Sep* = 9;
  Oct* = 10;
  Nov* = 11;
  Dec* = 12;

TYPE
  (* hh:mm:ss *)
  (* dd/mm/yy *)
  String* = ARRAY 9 OF CHAR;

VAR
  GetHMS*: PROCEDURE (VAR hrs, mins, secs: INTEGER);
  SetHMS*: PROCEDURE (hrs, mins, secs: INTEGER);
  GetDMY*: PROCEDURE (VAR dd, mm, yy: INTEGER);
  SetDMY*: PROCEDURE (dd, mm, yy: INTEGER);

PROCEDURE HMSToStr*(hrs, mins, secs: INTEGER; VAR s: String);

PROCEDURE DMYToStr*(dd, mm, yy: INTEGER; VAR s: String);

PROCEDURE StrToHMS*(s: String; VAR hrs, mins, secs: INTEGER): BOOLEAN;

PROCEDURE StrToDMY*(s: String; VAR dd, mm, yy: INTEGER): BOOLEAN;

END DateTime.
```

Description

The module RTC contains functions for setting and retrieving the date and time components of the Real Time Clock (RTC) and converting the date and time values to / from strings with validation.

yy is assumed to be in the range 00..99 representing the years 2000 to 2099.

StrToHMS expects the string to be in the 24 hr fixed format "hh:mm:ss". Values less than 10 must be padded with leading zeroes. The separator ":" can be any single character.

StrToHMS returns TRUE if hh is in the range 00..23, and mm and ss are in the range 00..59.

StrToDMY expects the string to be in the fixed format "dd/mm/yy". Values less than 10 must be padded with leading zeroes. The separator "/" can be any single character.

StrToDMY returns TRUE if yy is in the range 00..99, and the mm and dd values correctly follow the rules for the number of days in a month, accounting for leap years.

3.5 Error

Definition

```
DEFINITION MODULE Error;

IMPORT Put;

CONST
  (* Library codes *)
  input*    = 20; (* Input parameter has an unexpected value *)
  data*     = 21; (* Data has an unexpected value *)
  index*    = 22; (* Index out of bounds *)
  version*  = 23; (* Version check failed *)
  timeout*  = 24; (* Timeout value exceeded *)
  undefinedProc* = 25; (* Procedure variable not yet defined *)

TYPE
  String* = ARRAY maxMsgLen OF CHAR;
  ErrorMsgProc* = PROCEDURE (error: INTEGER; VAR msg: String);

VAR
  Msg*: ErrorMsgProc;

END Error.
```

Description

The module Error contains definitions for the system and library module error codes.

The procedure *Msg* takes an error code and returns a string containing the corresponding runtime error message or an empty string if it is a user-defined error.

A user-defined procedure can be assigned to Msg if user-defined error messages are required.

3.6 GPIO

Definition

```
DEFINITION MODULE GPIO;

CONST
  Mode_In* = 0;
  Mode_Out* = 1;
  Mode_AF* = 2;
  Mode_Analog* = 3;

  Push_Pull* = 0;
  Open_Drain* = 1;

  Speed_Low* = 0;
  Speed_Medium* = 1;
  Speed_High* = 2;
  Speed_VeryHigh* = 3;

  Floating* = 0;
  Pull_Up* = 1;
  Pull_Down* = 2;

  AF0* = 0;
  AF1* = 1;
  AF2* = 2;
  AF3* = 3;
  AF4* = 4;
  AF5* = 5;
  AF6* = 6;
  AF7* = 7;
  AF8* = 8;
  AF9* = 9;
  AF10* = 10;
  AF11* = 11;
  AF12* = 12;
  AF13* = 13;
  AF14* = 14;
  AF15* = 15;

TYPE
  PortConfiguration* = RECORD
    mode*: INTEGER;
    outputType*: INTEGER;
    speed*: INTEGER;
    resistors*: INTEGER;
    alternateFunction*: INTEGER
  END;

  Pin* = RECORD
    base*, no*: INTEGER
  END;

PROCEDURE* Map*(base, no: INTEGER; VAR pin: Pin);

PROCEDURE* Configure*(VAR pin: Pin; config: PortConfiguration);

PROCEDURE* Put*(pin: Pin; state: BOOLEAN);

PROCEDURE* Reset*(pin: Pin);

PROCEDURE* Set*(pin: Pin);

PROCEDURE* Get*(pin: Pin; VAR state: BOOLEAN);

END GPIO.
```

Description

The General Purpose Input/Output (GPIO) module contains functions to configure the hardware characteristics, and control the behaviour, of each of the GPIO pins.

Map associates a variable of type *Pin* with a port name and number.

Configure converts the symbolic configuration options in the PortConfiguration record to the numeric values required to set the configuration registers.

Get returns the current state of the pin. TRUE is high and FALSE is low.

Put sets the GPIO pin to high (state = TRUE) or clears it (state = FALSE).

Reset sets the GPIO pin to low.

Set sets the GPIO pin to high.

Example

The following extracts from the Astrobe Blinker example illustrates their use:

```
CONST
  PORTA = MCU.GPIOABase;

VAR
  LED: GPIO.Pin;
  config: GPIO.PortConfiguration;
  …
  …
  config.mode := GPIO.Mode_Out;
  config.speed := GPIO.Speed_High;
  config.outputType := GPIO.Push_Pull;
  config.resistors := GPIO.Floating;
  config.alternateFunction := GPIO.AF0
  …
  …
  (* Green LED is connected to GPIO port A5 *)
  GPIO.Map(PORTA, 5, LED);
  GPIO.Configure(LED, config);
  …
  …
  (* Switch the LED on *)
  GPIO.Set(LED);
```

Refer to the STM Reference manual and Datasheet for your microcontroller for an explanation of the configuration options.

3.7 Graphics

Definition

```
DEFINITION MODULE Graphics;

IMPORT Error;

TYPE
  DrawDotProc* = PROCEDURE (colour, x, y: INTEGER);

PROCEDURE Line*(colour, x0, y0, x1, y1: INTEGER);

PROCEDURE Circle*(colour, x0, y0, r: INTEGER);

PROCEDURE Ellipse*(colour, x0, y0, a, b: INTEGER);

PROCEDURE Init*(maxX, maxY: INTEGER; dd: DrawDotProc);

END Graphics.
```

Description

The module Graphics contains functions to draw lines, circles and ellipses on display devices that allow a dot to be drawn at specified x and y co-ordinates.

Init must be called before any other Graphics function. The *maxX and maxY* parameters specify the largest x and y co-ordinates at which a dot can be drawn. dd is a user-supplied procedure which draws a dot at the x and y co-ordinates e.g.

```
PROCEDURE DrawDot(colour, x, y: INTEGER);
BEGIN
   …
END DrawDot;

Init(132, 132, DrawDot);
```

References

Ref: Project Oberon - Wirth & Gutknecht, ACM Press 1992

3.8 I2C

Definition

```
DEFINITION MODULE I2C;

IMPORT Error, MCU, SYSTEM;

CONST
  I2C1* = 1;
  I2C2* = 2;
  …

TYPE
  ConfigurePinsProc* = PROCEDURE;

PROCEDURE WriteByte*(addr: BYTE; data: BYTE);

PROCEDURE Write*(addr: BYTE; data: ARRAY OF BYTE);

PROCEDURE ReadByte*(addr: BYTE; VAR data: BYTE);

PROCEDURE Read*(addr: BYTE; VAR data: ARRAY OF BYTE);

PROCEDURE Init*(bus: INTEGER; ConfigurePins: ConfigurePinsProc; freq: INTEGER);

END I2C.
```

Description

The module I2C contains functions to communicate with slave devices that are connected to the microcontroller's I2C bus. This enables you to write applications which can use a wide variety of external I2C standard parts, such as serial RAMs and EEPROMS, LCDs, and digital sensors such as accelerometers, compasses, temperature and pressure gauges etc.

Init must be called before any other I2C function.

- The *bus* parameter is *I2C1, I2C2* etc. depending on the capabilities of your target microcontroller.
- The *ConfigurePins* parameter is a user-supplied procedure which should configure the I2C function on the pins on the target microcontroller that the I2C device is connected to.
- The *freq* parameter is the I2C clock rate and must be 100000. It is provided to simplify porting to other versions of Astrobe which support more than one frequency.

The *addr* parameter is the 7-bit value that identifies the slave I2C device as specified in its datasheet. The *Read / Write* procedures shift the address value by one to allow for the read / write bit which is automatically set when reading, and cleared when writing.

The *data* parameter to *Read* and *Write* is declared as *ARRAY OF BYTE* so the actual parameter used can be of *any* data type. The number of bytes transferred is equal to the length of the array in bytes i.e. LEN(data).

See the I2C examples included with Astrobe for typical uses, example slave addresses and processor-specific configuration and initialisation code.

3.9 In

Definition

```
DEFINITION MODULE In;

IMPORT Error, Convert;

CONST
  (* Possible result values *)
  noError* = Convert.noError;
  overflow* = Convert.overflow;
  syntaxError* = Convert.syntaxError;

TYPE
  GetCharProc* = PROCEDURE (VAR ch: CHAR);

VAR
  result*: INTEGER;

PROCEDURE Init*(g: GetCharProc);

PROCEDURE Char*(VAR ch: CHAR);

PROCEDURE String*(VAR s: ARRAY OF CHAR);

PROCEDURE Int*(VAR n: INTEGER);

END In.
```

Description

The module In contains basic text formatting input functions.

Input can be redirected from any device that accepts ASCII character data by calling *Init* with the name of a procedure that inputs a single character to that device. When the system module Main is initialised, *Serial.GetCh* is the procedure passed to *In.Init* and input is directed from the serial port UART0.

In.Char reads the next character from the input device.

In.String skips all characters with a value less than an ASCII space (020X) and then reads all characters until LEN(s) characters have been read or a character with a value less than an ASCII space has been read. A terminating null character is appended to the array if it is not full.

In.Int reads the next string and attempts to convert it into an integer. The global value *result* is set to one of the possible result values depending on the success of the conversion.

3.10 LinkOptions

Definition

```
MODULE LinkOptions;

IMPORT SYSTEM;

VAR
  (* Startup parameters *)
  ConfigID*: INTEGER;
  HeapStart*: INTEGER;
  HeapLimit*: INTEGER;
  StackStart*: INTEGER;
  ResourceStart*: INTEGER;
  DataStart*: INTEGER;
  DataEnd*: INTEGER;
  CodeStart*: INTEGER;
  CodeEnd*: INTEGER;

END LinkOptions.
```

Description

The module LinkOptions contains variables which are initialised with data taken from the current configuration when the application was linked. Consequently, this data can be accessed by the application when it is running on the target MCU.

LinkOptions must be the first module to be loaded when an application is linked. This can normally be ensured by listing LinkOptions first in the IMPORT list of the Main module.

3.11 MCU

Definition

```
DEFINITION MODULE MCU;

IMPORT SYSTEM;

CONST
  (* see below *)

PROCEDURE* NVIC_EnableIRQ*(irqNo: INTEGER);

END MCU.
```

Description

Constants are included for the following peripheral and control registers:

Reset and Clock Control (RCC)
General Purpose Input/Output Ports (GPIO)
Timers (TIM)
Power Controller (PWR)
Real Time Clock (RTC)
Nested Vectored Interrupt Controller (NVIC)
Universal Synchronous Asynchronous Receiver Transmitter (USART)
Serial Peripheral Interface (SPI)
Inter-integrated Circuit (I2C)

Each of the STM32 microcontrollers includes a different set of these capabilities. See the corresponding source code module for details e.g.

```
AstrobeM3-v9.3\Lib\STM32L152\MCU.mod
```

NVIC_EnableIRQ is used in the initialisation of modules that implement interrupt handlers. It enables interrupts for the peripheral assigned to the IRQ position *irqNo*. These positions are enumerated in the NVIC chapter of the STM32 Reference manuals e.g. the irqNo value for timer TIM2 on the STM32L152 is 28.

The source code of *MCU* and *Main* can be modified to enable customisation of the startup process.

3.12 Main

Definition

```
DEFINITION MODULE Main;

IMPORT LinkOptions, In, MCU, Out, Serial, SYSTEM, Traps;

END Main.
```

Description

Main must be included in the list of imports in the main module of the application. The linker will report an error if Main was not loaded during the linking process.

When Main is initialised it executes the second-level startup code - memory mapping control, runtime error-trapping initialisation, clock configuration, phase-locked loop (PLL) setup etc.

Input/output is initialised to operate via a UART/USART by calling *In.Init* and *Out.Init* with procedures *Serial.GetCh* and *Serial.PutCh* respectively.

LinkOptions must be the first module to be loaded when an application is linked. This can normally be ensured by listing LinkOptions first in the IMPORT list of the Main module.

The source code of Main can be modified to enable user-customisation of the startup process.

3.13 Math - Mathematical Functions

Definition

```
DEFINITION MODULE Math;

IMPORT Error;

CONST

  (* Various useful constants *)
  pi*        = 3.14159266;
  twoDivPi*  = 0.63661977;
  fourDivPi* = 1.27323954;
  piDivTwo*  = 1.57079633;
  piDivFour* = 0.78539816;
  ln2*       = 0.69314718;

PROCEDURE Sqrt*(x: REAL): REAL;

PROCEDURE Ln*(x: REAL): REAL;

PROCEDURE Exp*(x: REAL): REAL;

PROCEDURE Sin*(x: REAL): REAL;

PROCEDURE Cos*(x: REAL): REAL;

PROCEDURE ArcTan*(x: REAL): REAL;

END Math.
```

Description

The module Math contains basic mathematical and trigonometric functions.

Sqrt returns the square root of *x.*

Ln is the natural logarithm (log to base *e*) of *x*.

Exp returns the value of the mathematical constant *e* to the power of *x*.

The parameter *x* for *Sin, Cos* and *ArcTan* is in radians (2 * pi radians = 360°).

3.14 MAU - Memory Allocation Unit

Definition

```
DEFINITION MODULE MAU;

IMPORT LinkOptions, SYSTEM;

TYPE
  Proc* = PROCEDURE (VAR p: INTEGER; T: INTEGER);

PROCEDURE New*(VAR p: INTEGER; T: INTEGER);

PROCEDURE Dispose*(VAR p: INTEGER; T: INTEGER);

PROCEDURE SetNew*(p: Proc);

PROCEDURE SetDispose*(p: Proc);

PROCEDURE Allocate*(VAR p: INTEGER; typeDesc: INTEGER);

PROCEDURE Deallocate*(VAR p: INTEGER; typeDesc: INTEGER);

END MAU.
```

Description

The module MAU contains the functions used by the system for dynamic variable memory allocation. MAU is dependent on the version of the compiler and must follow some specific conventions. It should not be replaced with a user-defined module and its interface definition must not be changed.

If a user module calls the Oberon NEW function to allocate dynamic memory to a pointer variable then *MAU.New* is automatically called and the MAU module is automatically imported as if you had added it to your import list. You should not call *MAU.New* directly.

MAU also contains some functions that can be called explicitly by a user module. If so, you must then explicitly include MAU in the module's IMPORT list.

MAU.New calls *Allocate* which assigns the required number of bytes of memory from the heap to the pointer variable.

MAU.Dispose calls *Deallocate* which can potentially be used to return dynamic memory that is no longer needed to the heap.

The standard versions of *Allocate* and *Deallocate* only make the memory available for later reuse if the block being deallocated is the most recent block to be allocated.

The standard versions of *Allocate* and *Deallocate* are included in the *Storage* library module so that you can modify them. *SetNew* can be used to replace the standard version of *Allocat*e, and *SetDispose* can be used to replace the standard version of *Deallocat*e with ones that you have written. See the documentation of the module Storage for more information.

3.15 Out

Definition

```
DEFINITION MODULE Out;

IMPORT Convert, Error;

TYPE
  PutCharProc* = PROCEDURE(ch: CHAR);

PROCEDURE Init*(p: PutCharProc);

PROCEDURE Char*(ch: CHAR);

PROCEDURE String*(s: ARRAY OF CHAR);

PROCEDURE Ln*();

PROCEDURE Int*(n, width: INTEGER);

PROCEDURE Hex*(n, width: INTEGER);

END Out.
```

Description

The module Out contains basic formatted text output functions. Output can be directed to any device that accepts ASCII character data by calling *Init* with the name of a procedure that outputs a single character to that device. When the system module Main is initialised, *Serial.PutCh* is the procedure passed to *Out.Init* and subsequent output is directed to the serial port UART0.

Procedure *Out.Ln* writes the carriage return / line feed pair of characters (*0DX, 0AX*)

Procedures with a *width* parameter output left-justified text. If the number of characters that are output is less than *width*, a sufficient number of blanks are output to make up the difference.

3.16 Put

Definition

```
DEFINITION MODULE Put;

IMPORT Error;

PROCEDURE* Init*();

PROCEDURE* EOS*(VAR s: ARRAY OF CHAR);

PROCEDURE* Ch*(ch: CHAR);

PROCEDURE* Str*(s: ARRAY OF CHAR);

PROCEDURE Int*(n: INTEGER);

END Put.
```

Description

The module Put contains string-handling helper functions used internally by the modules Convert and Reals.

3.17 Random

Definition

```
DEFINITION MODULE Random;

PROCEDURE Next*(range: INTEGER): INTEGER;

PROCEDURE* Seed*(value: INTEGER);

END Random.
```

Description

A pseudo-random number generator based on the example in *Programming in Oberon - Reiser & Wirth, ACM Press 1992*.

Next returns the next random number in a reproducible sequence.

Seed is called with the value 314159 when Random is initialised by the system.

Call *Random.Seed* with a different value if you want to initiate a different sequence of numbers.

3.18 Reals

Definition

```
DEFINITION MODULE Reals;

IMPORT Error, Put;

CONST
  (* Possible values for result *)
  noError* = 0;
  overflow* = 1;
  syntaxError* = 2;

PROCEDURE* Exponent*(x: REAL): INTEGER;

PROCEDURE* Mantissa*(x: REAL): REAL;

PROCEDURE Real*(m: REAL; e: INTEGER): REAL;

PROCEDURE* Ten*(e: INTEGER): REAL;

PROCEDURE StrToReal*(s: ARRAY OF CHAR; VAR x: REAL; VAR result: INTEGER);

PROCEDURE RealToStrE*(x: REAL; digits: INTEGER; VAR s: ARRAY OF CHAR);

PROCEDURE RealToStrF*(x: REAL; digits: INTEGER; VAR s: ARRAY OF CHAR);

END Reals.
```

Description

The module Reals contains functions to support operations using real numbers.

Exponent(r) returns the value of *exp,* and *Mantissa(r)* returns the normalised value of *r,* resulting from a call to the Oberon function *UNPK(r, exp). Real* constructs a real number from a normalised mantissa and exponent.

Ten returns the value 10.0^{e}

RealToStrE displays the real value in exponential notation, *RealToStrF* uses fixed point notation. *digits* is the number (1..7) of significant digits to use.

Examples

x	digits	RealToStrE	RealToStrF
0.0	1	0.0E+00	0.0
0.0	7	0.000000E+00	0.0
0.70710698	5	7.0711E-01	0.70711
0.70710698	6	7.07107E-01	0.707107
0.70710698	7	7.071070E-01	0.707107
0.999999	5	1.0000E+00	1.0
0.999999	6	9.99999E-01	0.999999
Reals.Ten(7)	7	1.000000E+07	10000000.0
Reals.Ten(20)	7	1.000000E+20	1.000000E+20

3.19 ResData

Definition

```
MODULE ResData;

IMPORT Error, LinkOptions, SYSTEM;

TYPE
  Resource* = POINTER TO RECORD
  END;

  DirEntry* = RECORD
    name*: Name;
    size*: INTEGER
  END;

PROCEDURE* Size*(r: Resource): INTEGER;

PROCEDURE* GetInt*(r: Resource; index: INTEGER; VAR data: INTEGER);

PROCEDURE* GetByte*(r: Resource; index: INTEGER; VAR data: BYTE);

PROCEDURE* GetChar*(r: Resource; index: INTEGER; VAR ch: CHAR);

PROCEDURE* GetIntArray*(r: Resource; index: INTEGER; count: INTEGER;
  VAR items: ARRAY OF INTEGER): INTEGER;

PROCEDURE* GetReal*(r: Resource; index: INTEGER; VAR data: REAL);

PROCEDURE* GetRealArray*(r: Resource; index: INTEGER; count: INTEGER;
  VAR items: ARRAY OF REAL): INTEGER;

PROCEDURE* Count*(): INTEGER;

PROCEDURE GetDirectory*(VAR list: ARRAY OF DirEntry);

PROCEDURE Open*(VAR r: Resource; name: ARRAY OF CHAR);

END ResData.
```

Description

The module ResData contains functions to access constant resource data items (e.g. fonts, bitmaps, input data etc.) that were appended to the executable by the Astrobe linker. See the section titled *Resource Data* in this document for more information.

name is the (case-sensitive) name of the module that had the same name as the linked resource file. If the name is longer than eight characters it is truncated to eight characters.

index is a zero-based 32-bit offset into the resource data.

Size returns the size of the resource data in bytes.

Count returns the number of named resources attached to the current application.

Example

If the first word in the binary data file *MyData.res* is a 32-bit integer with the value 10, the following code assigns 10 to the INTEGER variable *count.* The next ten words of resource data are then interpreted as REAL data and stored into the dynamically allocated array *weights*.

```
PROCEDURE GetData();
VAR
  r: ResData.Resource;
  count: INTEGER;
  weights: ARRAY OF REAL;
BEGIN
  ResData.Open(r, "MyData");
  IF ResData.Size(r) = 0 THEN
    (* Report error *)
  ELSE
    ResData.GetInt(r, 0, count);
    NEW(weights, count);
    ResData.GetRealArray(r, 1, count, weights)
    ...
```

3.20 RTC

Definition

```
DEFINITION MODULE RTC;

IMPORT MCU, SYSTEM;

PROCEDURE SetHMS*(hrs, mins, secs: INTEGER);

PROCEDURE SetDMY*(dd, mm, yy: INTEGER);

PROCEDURE GetHMS*(VAR hrs, mins, secs: INTEGER);

PROCEDURE GetDMY*(VAR dd, mm, yy: INTEGER);

END RTC.
```

Description

The module RTC contains functions for setting and retrieving the date and time components of the Real Time Clock (RTC).

yy is assumed to be in the range 00 .. 99 representing the years 2000 to 2099.

This module would not normally be directly imported. The same functions can be called indirectly from the DateTime module which also includes the functions to convert the date and time values to / from strings with validation.

3.21 Serial

Definition

```
DEFINITION MODULE Serial;

IMPORT SYSTEM, MCU, GPIO;

CONST
  USART2* = 2;
  USART3* = 3;

PROCEDURE* TxReady*(): BOOLEAN;

PROCEDURE* PutCh*(ch: CHAR);

PROCEDURE* RxReady*(): BOOLEAN;

PROCEDURE* GetCh*(VAR ch: CHAR);

PROCEDURE Init*(usartNo, baudRate: INTEGER);

END Serial.
```

Description

The module Serial contains functions for sending and receiving single ASCII characters via a USART serial port using polling.

Init is automatically called by *Main* to initialise USART2 with the baud rate set to 38,400 baud. The format is fixed at 8 bits, 1 stop bit and no parity.

The USART is selected by passing USART1 or USART2 as the usartNo parameter.

GetCh waits until a character is present in the UART receive buffer before returning it as a parameter. It should be used if a response is required before the program can proceed. Otherwise *RxReady* can be used to test if a character is available to be read before optionally calling *GetCh*. If *RxReady* returns FALSE other processing can be performed instead.

Similarly, *PutCh* waits until the UART transmit buffer is empty before sending the character. *TxReady* returns TRUE if the transmit buffer is empty and can be called before optionally calling *PutCh*. If *TxReady* returns FALSE other processing can be performed instead.

3.22 SPI

Definition

```
DEFINITION MODULE SPI;

IMPORT Error, MCU, SYSTEM;

CONST
  SPI0* = 0;
  SPI1* = 1;
  …

TYPE
  ConfigurePinsProc* = PROCEDURE;

PROCEDURE* Response*(): BYTE;

PROCEDURE* SendByte*(data: BYTE);

PROCEDURE* SendChar*(ch: CHAR);

PROCEDURE* SendData*(data: INTEGER);

PROCEDURE* Send*(buf: ARRAY OF BYTE);

PROCEDURE* ReceiveByte*(): BYTE;

PROCEDURE* ReceiveChar*(): CHAR;

PROCEDURE* ReceiveData*(): INTEGER;

PROCEDURE* Receive*(VAR buf: ARRAY OF BYTE);

PROCEDURE Init*(bus: INTEGER; ConfigurePins: ConfigurePinsProc);

END SPI.
```

Description

Serial Peripheral Interface (SPI) Functions used to communicate with slave devices that are connected to the microcontroller's SPI bus. This enables you to write applications which can use a wide variety of external SPI standard parts, such as LCD displays, SD cards, 7-segment LEDs, and digital sensors such as accelerometers, magnetometers, etc.

Init must be called before any other SPI function.

- The *bus* parameter depends on the SPI capabilities of the microcontroller in use (e.g. *SPI0, SPI1* etc.) It is used to specify which SPI bus subsequent calls will operate on.
- The *ConfigurePins* parameter is a user-supplied procedure which should configure the SPI function on the pins on the target microcontroller that the SPI device is connected to.

See the SPI examples included with Astrobe for typical uses, processor-specific configuration and initialisation code.

The *buf* parameters are declared as *ARRAY OF BYTE* so the actual parameters used can be of *any* data type.

3.23 Storage

Definition

```
DEFINITION MODULE Storage;

IMPORT LinkOptions, MAU, SYSTEM;

PROCEDURE Allocate*(VAR p: INTEGER; typeDesc: INTEGER);

PROCEDURE Deallocate*(VAR p: INTEGER; typeDesc: INTEGER);

PROCEDURE* HeapStart*(): INTEGER;

PROCEDURE* HeapPtr*(): INTEGER;

PROCEDURE* HeapUsed*(): INTEGER;

PROCEDURE* HeapAvailable*(): INTEGER;

PROCEDURE* StackStart*(): INTEGER;

PROCEDURE* StackPtr*(): INTEGER;

PROCEDURE* StackUsed*(): INTEGER;

PROCEDURE* StackAvailable*(): INTEGER;

END Storage.
```

Description

The module Storage contains a copy of the default dynamic memory allocation / deallocation procedures from the library module MAU that are invoked by the standard functions NEW / DISPOSE.

You can modify the source code of these functions to tailor their performance and efficiency to suit the requirements of your particular projects. There are additional benefits associated with being able to substitute your own functions e.g. you could add memory usage tracing features.

When the module Storage is initialised it calls

```
MAU.SetNew(Allocate);
MAU.SetDispose(Deallocate);
```

which replace MAU.Allocate and MAU.Deallocate with the functions Storage.Allocate and Storage.Deallocate respectively.

Storage also contains stack- and heap-monitoring procedures.

HeapStart returns the address of the start of the heap.

HeapPtr returns the address of the start of the block of memory that will be used for the next allocation from the heap.

HeapUsed returns the number of bytes or RAM currently used by all dynamic variables.

HeapAvailable returns the number of bytes of RAM currently free to be used for additional dynamic variables. By default this value is shared with the stack.

StackStart returns the address of the start of the stack.

StackPtr returns the address of the start of the block of memory that will be used for the next allocation from the stack.

StackUsed returns the number of bytes or RAM currently allocated to the stack.

StackAvailable returns the number of bytes of RAM currently free to be used for additional storage. By default this value is shared with the heap.

3.24 Strings

Definition

```
DEFINITION MODULE Strings;

IMPORT Error;

PROCEDURE* Length*(s: ARRAY OF CHAR): INTEGER;

PROCEDURE* Extract*(src: ARRAY OF CHAR; pos, n: INTEGER;
  VAR dest: ARRAY OF CHAR);

PROCEDURE* Copy*(src: ARRAY OF CHAR; VAR dest: ARRAY OF CHAR);

PROCEDURE* Append*(src: ARRAY OF CHAR; VAR dest: ARRAY OF CHAR);

PROCEDURE Pos*(pattern, s: ARRAY OF CHAR; pos: INTEGER): INTEGER;

PROCEDURE* Delete*(VAR s: ARRAY OF CHAR; pos, n: INTEGER);

PROCEDURE* Insert*(src: ARRAY OF CHAR; pos: INTEGER; VAR dest: ARRAY OF CHAR);

PROCEDURE Replace*(src: ARRAY OF CHAR; pos: INTEGER; VAR dest: ARRAY OF CHAR);

PROCEDURE* Cap*(VAR s: ARRAY OF CHAR);

END Strings.
```

Description

Contains general functions for processing strings i.e. string constants and arrays of characters terminated with the null character *0X*.

Length(s) returns the number of characters in s up to but excluding the first *0X*.

Extract(src, pos, n, dest) extracts a substring *dest* with n characters from position *pos (0 <= pos < Length(src))* in *src*.

Copy(src, dest) copies *Length(src)* characters to *dest*.

Append(s, dest) has the same effect as *Insert(s, Length(dest), dest).*

Pos(pat, s, pos) returns the position of the first occurrence of string pattern *pat* in s. Searching starts at position *pos*. If *pat* is not found, -1 is returned.

Delete(s, pos, n) deletes n characters from s starting at position *pos (0 <= pos < Length(s)).*

Insert(src, pos, dest) inserts the string *src* into the string *dest* at position pos (0 <= pos <= Length(dst)). If *pos = Length(dest), src* is appended to *dest*.

Replace(src, pos, dest) has the same effect as *Delete(dest, pos, Length(src))* followed by an *Insert(src, pos, dest).*

Cap(s) replaces each lower case letter within *s* by its upper case equivalent.

3.25 SYSTEM

Definition

```
DEFINITION MODULE SYSTEM;

PROCEDURE ADR*(variableName: <any type>): INTEGER;

PROCEDURE ALIGN*();

PROCEDURE BIT*(address, bitNo: INTEGER): BOOLEAN;

PROCEDURE CLZ*(data: INTEGER): INTEGER;

PROCEDURE COPY*(src, dest, nWords: INTEGER);

PROCEDURE EMIT*(instruction: INTEGER);

PROCEDURE EMITH*(instruction: INTEGER);

PROCEDURE EOR*(x, y: INTEGER): INTEGER;

PROCEDURE GET*(address: INTEGER; VAR v: <any basic type>);

PROCEDURE GET*(VAR address: INTEGER; VAR v: <any basic type>; inc: INTEGER);

PROCEDURE LDREG*(Rn, v: INTEGER);

PROCEDURE NULL*(x: <numeric type>): BOOLEAN;

PROCEDURE PUT*(address: INTEGER; x: <any basic type>);

PROCEDURE PUT*(VAR address: INTEGER; x: <any basic type>; inc: INTEGER);

PROCEDURE RBIT*(data: INTEGER): INTEGER;

PROCEDURE REG*(Rn: INTEGER): INTEGER;

PROCEDURE REV*(data: INTEGER): INTEGER;

PROCEDURE REV16*(data: INTEGER): INTEGER;

PROCEDURE REVSH*(data: INTEGER): INTEGER;

PROCEDURE SIZE*(typeName: <any type>): INTEGER;

PROCEDURE VAL*(typeName: <any type>; x: <any type>): typeName;

END SYSTEM.
```

Description

SYSTEM is a pseudo-module i.e. it contains no source code. Its functionality is implemented entirely within the compiler. Some of the functions allow parameters of any *basic* type i.e. INTEGER, SET, BOOLEAN etc. to be passed. Others allow parameters of *any* type. Generic functions of this type are normally not possible to write using the Oberon language.

The presence of SYSTEM in the IMPORT list of a module indicates that the module is implementation-dependent.

ADR returns the absolute address of the given variable.

BIT returns TRUE if the specified bit of the contents of the given address is equal to 1, otherwise FALSE.

COPY copies *nWords* consecutive words from *src* to *dest*, where *src*, *dest* and *nWords* are all INTEGERs. *src* and *dest* are absolute memory addresses and can be obtained from variable names using the SYSTEM.ADR function.

GET loads the value of the word or byte at absolute memory location *address* into variable *v*. The size of the variable *v* determines whether a *load register byte* (LDRB) or *word* (LDR) instruction is used to perform the transfer.

PUT stores the value of *x* into the word or byte at absolute memory location *address*. The size of *x* determines whether a *store register byte* (STRB) or *word* (STR) instruction is used to perform the transfer.

An optional integer constant parameter *inc* can be used with *GET* and *PUT* to automatically increment / decrement the value of *address*. If *inc* is in the range 1..255 the address is incremented after the value is stored. If *inc* is in the range -1..-255 the address is decremented before the value is stored. Typical values for *inc* are 1 for byte accesses and 4 for word accesses.

NOTE: If *address* is a global record field or array element and *inc* is non-zero the compiler will report a *read-only* error.

SIZE returns the number of bytes used by a variable of the given type.

Astrobe ARM Cortex-M Extensions:

Extensions to the standard Oberon SYSTEM features are provided to give low-level access to ARM Cortex-M features that might otherwise require the use of assembly language.

ALIGN inserts a NOP instruction if necessary at the current code location to ensure that the next instruction is aligned on a word boundary.

CLZ counts the number of leading zeroes in an integer value.

EMIT inserts the 32-bit value of an ARM Thumb-2 instruction at the current code location. Examples of its use can be seen in the source code of the *Traps* and *IAP* library modules.

EMITH inserts the 16-bit value of an ARM Thumb instruction at the current code location. A compilation error results if the value is not a valid 16-bit positive integer.

EOR (x, y) performs a bitwise Exclusive OR of the two integer values.

LDREG (Rn, v) stores the value v in CPU register Rn where Rn is a constant in the range 0..15.

NULL returns true if x is negative or positive INTEGER or REAL zero.

RBIT, REV, REV16 and REVSH are bit and byte-ordering functions. Each generates the single Thumb-2 instruction with the same name:

- *RBIT* reverses the bit order of the integer value.
- *REV* reverses the byte order of the integer value.

- *REV16* reverses the byte order in each 16-bit halfword of the integer value.

- *REVSH* reverses the byte order in the lower 16-bit halfword of the integer value and sign extends the result to 32 bits.

REG (Rn) returns the current value of CPU register Rn where Rn is a constant in the range 0..15. e.g. SYSTEM.REG(13) returns the current value of the stack pointer. Use a local variable in a non-leaf procedure to store the returned result to avoid this register value being overwritten when it is assigned. Refer to the Disassembler listing to check that the code generated is what you expected.

VAL is a *type-transfer* / *typecast* mechanism. It should be used with extreme care as it effectively bypasses any type-safety checks except to ensure that the sizes of each type are the same. It allows the value of *x* (which can be of any type) to be interpreted as if it were declared as type *typeName*. No value conversion takes place.

3.26 Timers

Definition

```
DEFINITION MODULE Timers;

IMPORT Error, MCU, SYSTEM;

CONST
  uSecs* = 1000000;
  MSecs* = 1000;

  TIM0* = 0;
  TIM1* = 1;
  ...

TYPE
  Timer* = RECORD
    base*, multiplier*, units*: INTEGER
  END;

PROCEDURE* Delay*(timer: Timer; delay: INTEGER);

PROCEDURE* Init*(VAR timer: Timer; timerNo, units: INTEGER);

PROCEDURE* Start*(timer: Timer);

PROCEDURE* Stop*(timer: Timer);

PROCEDURE* Elapsed*(timer: Timer): INTEGER;

END Timer.
```

Description

The module Timers contains functions for timed delays and for measuring elapsed execution time. Several different timers can be used in the same application. The *timer* parameter determines which one is used.

Init must be called before the first call to each different timer. *timerNo* is used to associate one of the timers (e.g. *TIM2*) to the *timer* variable*. units* can be either *Timers.MSecs* or *Timers.uSec.* The units specified in *Init* are used for that timer for all subsequent calls. The units can be changed by calling *Init* again.

NOTE: Check the number of timers that are available and their capabilities in your microcontroller reference manual. For example, not all timers are able to handle microsecond intervals.

Examples are:

```
Init(usecTimer, Timers.TIM2, Timers.uSecs);
Init(msecTimer, Timers.TIM3, Timers.MSecs);
```

Consequently, both of these calls would both result in a 1-second delay:

```
Delay(usecTimer, 1000000);
Delay(msecTimer, 1000);
```

The delay functions reset the timer and then execute a continuous loop until the measured elapsed time exceeds the given *delay*.

Elapsed returns the time that passed between the two most recent calls of *Start* and *Stop*.

Because the delay functions reset the timer they should not be called in sections of code that are being timed by the same timer.

3.27 Traps

Definition

```
DEFINITION MODULE Traps;

IMPORT LinkOptions, Error, Out, ResData, SYSTEM;

CONST
  MinUserTrap* = 200;
  MaxTrap*     = 255;
  TraceDepth   = 16;

TYPE
  InterruptHandler* = PROCEDURE();
  Name* = ARRAY 16 OF CHAR;
  ID* = RECORD
    addr*, lineNo*: INTEGER;
  END;
  Trace* = RECORD
    id*: ARRAY TraceDepth OF ID;
    errorCode*, count*: INTEGER
  END;

  UserHandler* = PROCEDURE(trace: Trace);

VAR
  trace*: Trace;

PROCEDURE* ShowRegs*(b: BOOLEAN);

PROCEDURE GetName*(target: INTEGER; VAR modName, procName: Name);

PROCEDURE* Length*(s: ARRAY OF CHAR): INTEGER;

PROCEDURE* Assign*(addr: INTEGER; p: InterruptHandler);

PROCEDURE* SetUserHandler*(code: INTEGER; p: UserHandler);

PROCEDURE Init*;

END Traps.
```

Description

The module Traps Implements default interrupt handlers for software interrupts and the standard internal Cortex-M exceptions: e.g. *NMI* and *Hard Fault*. The default handlers are installed when *Traps.Init* is called from the *Main.Init* function at startup time.

The supervisor call (*SVC*) instruction handler is invoked whenever Astrobe executes a statement which results in a runtime error (e.g. array index out of range, division by zero etc.)
Control also passes to the handler if an ASSERT statement is executed with a parameter which equates to FALSE.

Traps.Assign is used to assign an interrupt handler to the related interrupt vector.

Traps.GetName identifies the name of the module and the procedure within it which contains the instruction with the address *Target*.

Traps.Length returns the number of characters contained in the string *s.*

Traps.ShowRegs determines whether or not register values are displayed when the trap handler is executed.

Traps.SetUserHandler allows a user-defined procedure that has been written to handle a runtime error differently from the default behaviour, to be assigned to a specific assertion code in the range 200 .. 255. An example application, called *UserTraps*, is supplied with Astrobe to demonstrate how this can be done.

You can modify the source code of *Traps* to enable further user-customisation of the interrupt handling process if required.

See the *Interrupt Handlers* section above and the *Runtime Errors* section below for more details.

4 Debugging

4.1 Runtime Error Codes

The error codes assigned to runtime errors and assertions detected by Oberon are:

Code	Reason
1	Index out of bounds
2	Type test failure
3	Source and destination arrays are not the same length
4	Invalid value in case statement
5	Attempt to call a NIL procedure variable
6	String too long or destination string too short
7	Integer division by zero or negative divisor
8, 9, 10	FPU assertions
11	Reserved
12	Attempt to dispose a NIL pointer
13..19	Reserved
20..25	Library assertions – see the Error module for definitions
26..99	Reserved
100..199	User-defined assertions
200..255	User-defined assertions with customisable trap handlers

4.2 User-defined Assertions

You can use the Oberon ASSERT function to trap an application-specific error e.g. to detect impending stack overflow:

```
ASSERT(Storage.StackAvailable < minRequired, 130)
```

where `minRequired` is a user-defined value.

User-defined assertions should use error codes in the range 100 – 255 to distinguish them from Runtime and Library errors.

Error codes 100 – 199 will display error information in the same way as the Library errors.

Error codes 200 – 255 can be used if you want to handle the error in a different way. An example application, called *UserTraps*, is supplied with Astrobe to demonstrate how this can be done.

4.3 Reporting Runtime Errors

The above runtime, library and programmer-defined error conditions and assertions result in the execution of a Cortex-M supervisor call instruction (SVC) which calls a default trap handler in the Astrobe library module *Traps*.

The trap handler reports:

- an error code or message describing what type of error it is
- the name of the module and procedure that was being executed
- the address of the instruction which caused the error
- the line number of the corresponding statement in the source code
- the values of the registers which are automatically saved at the time of the runtime error or assertion failure

If the *Stack Trace* option on the Astrobe Configuration dialog was enabled when the module was compiled, the details of the sequence of procedure calls that led to the error are included:



If the procedure call Traps.ShowRegs(FALSE) is made before the trap occurs the display of register values is suppressed. This is useful if the display only has a few lines and cannot show all of the information without scrolling.

The error messages that are displayed are defined in the module *Error*. If there is no message corresponding to the error code, the error code is displayed instead.
The information is reported using the standard IO functions exported by the Astrobe *Out* module. By default the messages will appear on a serial terminal connected to UART0. The trap handler then processes an infinite loop until the system is reset.

You can modify the source code of *Traps* to allow customisation of the trap-handling process.

When debugging your program, you can use the register values in conjunction with the assembly listing of the module or application to help identify the values of variables at the time of failure.

4.4 Diagnosing Runtime Errors

When a runtime error occurs or an assertion fails, use the module name and line number information reported by the trap handler to identify the source of the error.

- Open the source code of the named module in the editor
- Use the *Search > Goto* command to locate the actual source line by its line number.

4.5 Diagnosing System Exceptions

Traps caused by runtime errors or assertion failures which result in Supervisor Calls (SVC) are easy to locate as they give you the module name and line number of the offending line of source code. Hardware-related and other system exceptions are more difficult to locate as they only give you the module name and the address of the instruction that failed. Fortunately they are much rarer than runtime errors.

The type of Cortex-M hardware system exceptions handled by the Astrobe *Traps* module can include the following:

- NMI
- Hard Fault
- Memory Manager
- Bus Fault
- Usage Fault

Refer to the *ARM v7-M Architecture Reference Manual* which can be downloaded from the ARM website, for details of the possible causes of these exceptions.

If the exception is not caused by a secondary effect, and your edition of Astrobe includes the Module and Application Disassemblers it is usually possible to identify the line of code in your application which generated the offending instruction. To do this you need to have:

- The runtime error message displayed when your application terminated. This will give you the module name and exception address.
- The map file for the main module (*<ModuleName>.map*) which was created when you linked / built the application. The start address of the module is listed in the *Code Address* column of the map file.
- A Module Disassembler listing (*Project > Disassemble Module*) or an Application Disassembler listing (*Project > Disassemble Application*) of the problem module.

4.5.1 Using the Module Disassembler Listing:



You can calculate the offset and find the corresponding line of code in the disassembly listing using the following formula:

offset = exception address – start address – n

where n is 8 (4 for Cortex-M0). It is subtracted because the Cortex-M program counter is ahead of the current instruction by that many bytes. If you look in the disassembler listing for the instruction with the same offset you will see the accompanying Oberon source line which generated that instruction.

4.5.2 Using the Application Disassembler Listing:



You can calculate the offset and find the corresponding line of assembler code with that offset in the disassembly listing using the following formula:

offset = exception address – code start address

where the addresses are hexadecimal numbers and *code start address* is the first *Code Range* entry on the Astrobe Configuration dialog.

The heading of that block of assembly instructions will show the name of the module and procedure where the instruction is located.

5 Compile, Link and Build Commands

Separate command-line programs for the Oberon Cortex-M Compiler, Builder and Linker which correspond to the built-in compile, build and link commands in the IDE are included with the Professional Edition of Astrobe.

The separate compiler, builder and linker can be used with automatic 'build' tools, DOS-batch commands etc. These are useful for handling a regular series of compilations and links when building multiple configurations, multiple targets etc. They can also be useful when recompiling a number of modules after changing the interface of a low-level imported module or upgrading to a newer version of Astrobe.

All of the commands have two required parameters.

```
AstrobeCompile <configfile>.ini [<path>]<ModuleName>.mod

AstrobeBuild <configfile>.ini [<path>]<MainModuleName>.mod

AstrobeLink <configfile>.ini [<path>]<MainModuleName>.mod
```

MainModuleName is the filename of the main module being compiled or linked.

configfile is the name of the configuration file containing the options to use.

5.1 Examples

```
AstrobeCompile  D:\AstrobeM3-v9.3\Configs\STM32L152.ini  Lists.mod

AstrobeBuild  D:\AstrobeM3-v9.3\Configs\STM32L152.ini  Blinker.mod

AstrobeLink  D:\AstrobeM3-v9.3\Configs\STM32L152.ini  Blinker.mod
```

5.2 Command Return Codes

If the command executes without any compiler or linker errors it returns zero otherwise it returns 1. Examples of DOS batch scripts, for use with Astrobe for Cortex-M3, which use these return values are:

```
REM
REM Rebuild General Library
REM
SET cfg=%AstrobeM3%\configs\STM32L152.ini
SET compile="C:\Program Files\AstrobeM3\AstrobeCompile.exe"
REM
cd %AstrobeM3%\Lib\General
del *.arm
del *.smb
%compile% %cfg% Math.Mod
if errorlevel 1 goto ErrorExit
%compile% %cfg% Random.Mod
if errorlevel 1 goto ErrorExit
%compile% %cfg% Put.Mod
if errorlevel 1 goto ErrorExit
%compile% %cfg% Convert.Mod
if errorlevel 1 goto ErrorExit
%compile% %cfg% In.Mod
if errorlevel 1 goto ErrorExit
%compile% %cfg% Out.Mod
if errorlevel 1 goto ErrorExit
%compile% %cfg% LinkOptions.Mod
if errorlevel 1 goto ErrorExit
%compile% %cfg% ResData.Mod
if errorlevel 1 goto ErrorExit
%compile% %cfg% Traps.Mod
if errorlevel 1 goto ErrorExit
%compile% %cfg% Reals.Mod
if errorlevel 1 goto ErrorExit
%compile% %cfg% Strings.Mod
if errorlevel 1 goto ErrorExit
echo No errors detected
goto OK
:ErrorExit
echo Errors detected
:OK
pause
```

```
REM
REM Rebuild General Library
REM
SET cfg=%AstrobeM3%\configs\STM32L152.ini
SET build="C:\Program Files\AstrobeM3\AstrobeBuild.exe"
REM
cd %AstrobeM3%\Lib\General
del *.arm
del *.smb
%build% %cfg% Build.Mod
if errorlevel 1 goto ErrorExit
echo No errors detected
goto OK
:ErrorExit
echo Errors detected
:OK
pause
```